

NAG Library Routine Document

F08JFF (DSTERF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08JFF (DSTERF) computes all the eigenvalues of a real symmetric tridiagonal matrix.

2 Specification

```
SUBROUTINE F08JFF (N, D, E, INFO)
INTEGER N, INFO
REAL (KIND=nag_wp) D(*), E(*)
```

The routine may be called by its LAPACK name *dsterf*.

3 Description

F08JFF (DSTERF) computes all the eigenvalues of a real symmetric tridiagonal matrix, using a square-root-free variant of the *QR* algorithm.

The routine uses an explicit shift, and, like F08JEF (DSTEQR), switches between the *QR* and *QL* variants in order to handle graded matrices effectively (see Greenbaum and Dongarra (1980)).

4 References

Greenbaum A and Dongarra J J (1980) Experiments with QR/QL methods for the symmetric triangular eigenproblem *LAPACK Working Note No. 17 (Technical Report CS-89-92)* University of Tennessee, Knoxville <http://www.netlib.org/lapack/lawnpdf/lawn17.pdf>

Parlett B N (1998) *The Symmetric Eigenvalue Problem* SIAM, Philadelphia

5 Parameters

1: N – INTEGER	<i>Input</i>
<i>On entry:</i> n, the order of the matrix T.	
<i>Constraint:</i> N ≥ 0.	
2: D(*) – REAL (KIND=nag_wp) array	<i>Input/Output</i>
<i>Note:</i> the dimension of the array D must be at least max(1,N).	
<i>On entry:</i> the diagonal elements of the tridiagonal matrix T.	
<i>On exit:</i> the n eigenvalues in ascending order, unless INFO > 0 (in which case see Section 6).	
3: E(*) – REAL (KIND=nag_wp) array	<i>Input/Output</i>
<i>Note:</i> the dimension of the array E must be at least max(1,N – 1).	
<i>On entry:</i> the off-diagonal elements of the tridiagonal matrix T.	
<i>On exit:</i> E is overwritten.	
4: INFO – INTEGER	<i>Output</i>
<i>On exit:</i> INFO = 0 unless the routine detects an error (see Section 6).	

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

The algorithm has failed to find all the eigenvalues after a total of $30 \times N$ iterations. If INFO = i , then on exit i elements of E have not converged to zero.

7 Accuracy

The computed eigenvalues are exact for a nearby matrix $(T + E)$, where

$$\|E\|_2 = O(\epsilon)\|T\|_2,$$

and ϵ is the *machine precision*.

If λ_i is an exact eigenvalue and $\tilde{\lambda}_i$ is the corresponding computed value, then

$$|\tilde{\lambda}_i - \lambda_i| \leq c(n)\epsilon\|T\|_2,$$

where $c(n)$ is a modestly increasing function of n .

8 Parallelism and Performance

Not applicable.

9 Further Comments

The total number of floating-point operations is typically about $14n^2$, but depends on how rapidly the algorithm converges. The operations are all performed in scalar mode.

There is no complex analogue of this routine.

10 Example

This example computes all the eigenvalues of the symmetric tridiagonal matrix T , where

$$T = \begin{pmatrix} -6.99 & -0.44 & 0.00 & 0.00 \\ -0.44 & 7.92 & -2.63 & 0.00 \\ 0.00 & -2.63 & 2.34 & -1.18 \\ 0.00 & 0.00 & -1.18 & 0.32 \end{pmatrix}.$$

10.1 Program Text

```
Program f08jffe
!
! F08JFF Example Program Text
!
! Mark 25 Release. NAG Copyright 2014.
!
! .. Use Statements ..
Use nag_library, Only: dstterf, nag_wp
!
! .. Implicit None Statement ..
Implicit None
!
! .. Parameters ..
Integer, Parameter :: nin = 5, nout = 6
!
! .. Local Scalars ..
Integer :: info, n
!
! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: d(:), e(:)
```

```

!     .. Executable Statements ..
Write (nout,*) 'F08JFF Example Program Results'
! Skip heading in data file
Read (nin,*)
Read (nin,*) n

Allocate (d(n),e(n-1))

! Read T from data file

Read (nin,*) d(1:n)
Read (nin,*) e(1:n-1)

! Calculate the eigenvalues of T
! The NAG name equivalent of dsterf is f08jff
Call dsterf(n,d,e,info)

Write (nout,*)
If (info>0) Then
    Write (nout,*) 'Failure to converge.'
Else
    Write (nout,*) 'Eigenvalues'
    Write (nout,99999) d(1:n)
End If

99999 Format (3X,(9F8.4))
End Program f08jffe

```

10.2 Program Data

```

F08JFF Example Program Data
 4                               :Value of N
-6.99    7.92    2.34    0.32
-0.44   -2.63   -1.18      :End of matrix T

```

10.3 Program Results

```

F08JFF Example Program Results

Eigenvalues
 -7.0037 -0.4059  2.0028  8.9968

```
